
日本ソフトウェア科学会第 38 回大会 (2021 年度) 講演論文集

解集合プログラミングに基づく組合せ遷移ソルバーの
実装方式に関する考察

山田 悠也　湊 真一　番原 睦則

組合せ遷移問題とは，基となる組合せ問題とその 2つの実行可能解が与えられたとき，一方の実行可能解から他方の
実行可能解へ，遷移制約を満たしつつ，実行可能解のみを経由して到達できるかを判定する問題である．計算困難性
について，基の組合せ問題が NP 完全であるとき，その遷移問題の多くは PSPACE 完全になることが知られてい
る．本稿では，解集合プログラミング (ASP) を用いた組合せ遷移問題の解法として，有界組合せ遷移を提案し，そ
のソルバーの実装方法について述べる．有界組合せ遷移は，組合せ遷移問題に対して，制限された長さの (すなわち
有界の) 遷移系列を論理プログラムとして記号的に表現し，その論理プログラムを ASP システムで実行することに
より，到達可能性の検査を行う．有界組合せ遷移の有効性を評価するために，代表的な組合せ遷移問題の 1つである
k 彩色遷移問題をベンチマークとして，提案ソルバーの性能評価を行った．

1 はじめに
組合せ遷移問題 (Combinatorial Reconfiguration

Problems; CoRe [12] [14] [16]) とは，基となる組合せ
問題とその 2 つの実行可能解が与えられたとき，一
方の実行可能解から他方の実行可能解へ，遷移制約を
満たしつつ，実行可能解のみを経由して到達できるか
どうかを判定する問題である．
計算困難性について，基の組合せ問題が NP 完全
であるとき，その遷移問題の多くは PSPACE 完全に
なることが知られている．代表的なものとしては，命
題論理の充足可能性判定問題 (SAT)，集合被覆問題，
独立点集合，グラフ点彩色問題を基とする組合せ遷移
問題などがある [5] [6] [10] [12]．
組合せ遷移問題の研究は，理論計算機科学の分野を
中心に，最近 10年で急速に発展し，理論的な基盤が
整備されつつある．また，実社会においても，配電網

の切替えなど持続的システムへの実用的応用が期待
されている．しかしながら，一般の組合せ遷移問題を
解く汎用ソルバーのアルゴリズム，ヒューリスティッ
クスなどの実践的な研究開発はまだ始まったばかりで
ある．著者らの知る限り，実用的な汎用ソルバーはま
だ存在していない．
解集合プログラミング (Answer Set Programming;

ASP [2] [9] [11] [13]) は，論理プログラミングから派生
した宣言的プログラミングパラダイムである．ASP

言語は一階論理に基づく知識表現言語の一種である．
ASP システムは安定モデル意味論 [9]に基づく解集
合を計算するシステムである．近年，SAT の技術を
応用した高速な ASP システムが開発され，プランニ
ング，有界モデル検査，システム生物学など様々な分
野への実用的応用が急速に拡大している [7].

組合せ遷移問題に対して ASP を用いる利点とし
ては，ASP 言語の高い表現力により，記号制約を簡
潔に記述できる点，インクリメンタル ASP 解法によ
り，学習節を再利用した効率的な解探索が可能である
点，探索ヒューリスティックスを簡単にカスタマイズ
できる点，などが挙げられる．
本稿では，解集合プログラミング (ASP)を用いた
組合せ遷移問題の解法として，有界組合せ遷移を提案



し，そのソルバーの実装方法について述べる．有界組
合せ遷移は，SAT 技術を用いたモデル検査手法の 1

つである有界モデル検査 [1] [3] [4] を組合せ遷移に応
用した解法といえる．
有界組合せ遷移は，組合せ遷移問題に対して，制限
された長さの (すなわち有界の)遷移系列を論理プロ
グラムとして記号的に表現し，その論理プログラム
を ASP システムで実行することにより，到達可能性
の検査を行う．ASP システムが充足可能と判定した
場合，制限された長さの到達可能な遷移系列が存在
することを意味する．逆に充足不能と判定した場合，
制限された長さでは到達不能であることを意味する.

この場合，遷移系列の長さを増加させた論理プログラ
ムを構成し，再び ASP システムによる実行を繰り返
す．有界組合せ遷移は，到達可能な遷移系列を探すだ
けで，到達不能の証明は行わない不完全な手続きであ
るが，検査すべき遷移系列の長さに上限が存在する場
合には完全な手続きとなる．
有界組合せ遷移の有効性を評価するために，代表
的な組合せ遷移問題の 1つである k 彩色遷移問題を
ベンチマークとして，提案ソルバーの性能評価を行っ
た．その結果，インクリメンタル ASP 解法を用いた
実装方法がより多くの問題を高速に解くことができ
ることを確認した．

2 組合せ遷移問題と k彩色遷移問題
組合せ遷移問題 (Combinatorial Reconfiguration

Problems; CoRe [12] [14] [16]) は，基となる組合せ問
題の実行可能解が形成する解空間グラフを用いて定
義される．解空間グラフの頂点は実行可能解を表し，
辺は遷移制約に関する実行可能解の間の隣接関係を
表す．つまり，組合せ問題が実行可能解が存在するか
を判定するのに対し，組合せ遷移問題は，解空間グラ
フにおいて 2 つの頂点間の到達可能性を判定するこ
とが目的となる．
基となる組合せ問題の例として，色数 k のグラフ
点彩色問題を考える．この問題は，与えられた有限無
向グラフについて，隣接する頂点が同色にならないよ
うに各頂点を彩色するとき，グラフが k 色以下で彩
色可能かどうかを判定する問題である．

図 1 色数 k = 3 のグラフ点彩色問題の入力例 (左) と実

行可能解 (右)

グラフ点彩色問題を基とする組合せ遷移問題は，
k 彩色遷移問題 (k-coloring Reconfiguration Prob-

lems) ともよばれ盛んに研究されている [5] [6]．k 彩
色遷移問題は，色数 k のグラフ点彩色問題と 2つの
実行可能解が与えられたとき，一方の実行可能解から
他方の実行可能解へ，1回の遷移において色が変化す
る頂点はただ 1つのみという遷移制約を満たしつつ，
実行可能解のみを経由して到達できるかを判定する
問題である．k彩色遷移問題では，グラフの形や色数
に制限を加えることで多項式時間で解ける問題が存
在するが，一般に k ≥ 4 において PSPACE 完全で
あることが知られている．
図 1に，色数 k = 3のグラフ点彩色問題の例とその
実行可能解を示す．図 2は，到達可能な k = 4彩色遷
移問題の例である．この例は，スタート状態 (t = 0)

からゴール状態 (t = 3)へ，3ステップの遷移で到達
している．また，各状態 tはグラフ点彩色問題の制約
を満たしており，1回の遷移においてちょうど 1個の
頂点の色が変化していることがわかる．

3 解集合プログラミング
解集合プログラミング (Answer Set Programming;

ASP [2] [9] [11] [13]) は，演繹データベース，否定を含
む論理プログラミング，非単調推論，制約充足 (特に，
SAT)を起源にもつ，宣言的プログラミングパラダイ
ムである．ASP の言語は，一般拡張選言プログラム
に基づいている．本稿では説明の簡略化のため，その
サブクラスである標準論理プログラムについて説明
する．以降，標準論理プログラムを単に論理プログラ
ムとよぶ．
論理プログラムは，以下の形式をしたルールの有限

図 2 到達可能な k = 4 彩色遷移問題の例

集合である．
a0 ← a1, . . . , am,∼am+1, . . . ,∼an

ここで，0 ≤ m ≤ n であり，各 ai はアトム，∼はデ
フォルトの否定 †1，“,” は連言を表す．← の左側を
ヘッド，右側をボディとよぶ．ルールの直観的な意味
は，「a1, . . . , am がすべて成り立ち，am+1, . . . , an の
それぞれが成り立たないならば，a0 が成り立つ」で
ある．ボディが空のルール (すなわち a0 ←)をファク
トとよび，←を省略してよい．
ヘッドが空のルールを一貫性制約とよぶ．

← a1, . . . , am,∼am+1, . . . ,∼an

直観的には，← a1 は，a1 ではないという禁止を表
し，← ∼a1 は，a1 でなければならないという強制を
表す．また，← a1,∼a2は，a1が成り立つならば，a2

が成り立つことを意味する．
ASP 言語は，組合せ問題を解くための便利な拡張
構文を備えている．その代表的なものが選択子と個数
制約である．例えば，選択子 {a1; . . . ; an} をファク
トとして書くと，アトム集合 {a1, . . . , an} の任意の
部分集合が成り立つことを意味する．個数制約は選
択子の両端に選択可能な個数の上下限を付けたもの
である．例えば，lb {a1; . . . ; an} ub← Body と書く
と，Body が成り立つならば，a1, . . . , an のうち，lb

個以上 ub個以下が成り立つことを意味する．
ASP システムは，与えられた論理プログラムか
ら，安定モデル意味論 [9] に基づく解集合を計算す
るソフトウェアである．ASP システムの多くは，変
数を含む論理プログラムを変数を含まない論理プロ

†1 “失敗による否定” ともよばれる．述語論理で定義さ
れる否定 (¬) とは意味が異なる．

表 1 論理プログラムとソースコードの対応

論理プログラム ← , ; ∼

ソースコード :- , ; not

グラムに基礎化したのち，基礎ソルバーを用いて解
集合を計算する．本稿で使用する高速 ASP システム
clingo †2 は，基礎化のためのグラウンダー gringo と
基礎ソルバー clasp をシームレスに結合したものであ
る．以降で示す論理プログラムのソースコードはすべ
て clingo 言語で書かれており，表記上の対応につい
ては表 1の通りである．
ASP を用いた基本的な問題解法は，最初に解きた
い問題を論理プログラムとして表現 (ASP符号化)す
る．つぎに ASP システムを用いて論理プログラムの
解集合を計算する．最後に解集合を解釈してもとの問
題の解を得る，という 3 つのステップからなる．以
下，グラフ点彩色問題を例にとって，各ステップごと
に説明する．
コード 1 に，図 1 のグラフを ASP のファクト形
式で表したもの示す．アトム n/1は頂点数，e/1は辺
数，node/1は頂点，edge/2は辺を表している．ピリ
オド (“.”)はルールの終わりを表す終端記号である．
グラフ点彩色問題を解く論理プログラムを，コー
ド 2に示す．コード中の cは色数を表す定数であり，
実際の値は実行時に clingoのオプションとして与えら
れる．1行目のアトム col/1は色を表し，col(1..c).
は col(1)., col(2)., . . . , col(c).と書くことに等
しい．2行目のルールは，頂点 Xが色 Cで塗られるこ
とを意味するアトム color(X,C)を導入し，個数制約

†2 https://potassco.org/



n(4). e(4).

node(1). node(2). node(3). node(4).

edge(1,2). edge(2,3). edge(2,4). edge(3,4).

コード 1 図 1 のグラフの ASP ファクト表現

(graph.lp)

1 col(1..c).

2 1 { color(X,C): col(C) } 1 :- node(X).

3 :- not { color(X,C); color(Y,C) } 1, edge(X,Y), col(C).

コード 2 色数 c のグラフ点彩色問題を解く論理プログラ

ム (color.lp)

% clingo -c c=3 graph.lp color.lp

clingo version 5.4.0

Reading from graph.lp ...

Solving...

Answer: 1

color(1,1) color(2,3) color(3,1) color(4,2)...

SATISFIABLE

Models : 1+

Calls : 1

CPU Time : 0.002s

コード 3 clingo の実行例

を用いて「各頂点は 1 つの色で塗られる」という制
約を表している．3行目のルールは，一貫性制約と個
数制約を使って「辺で結ばれた頂点 Xと Yが同じ色 C

で塗られることはない」という制約を表している．
ASP システムは，コード 1のファクトとコード 2

の論理プログラムから解集合を計算する．コード 3

に clingo の実行例を示す．色 1 を赤，色 2 を青，
色 3 を黄とすると，得られた解集合 { color(1,1),
color(2,3), color(3,1), color(4,2)} は，図 1の
右側の彩色を表す．
論理プログラムの基礎化について，もう少し詳しく
説明する．clingo は，グラウンダー gringo を用いて
コード 1のファクトとコード 2の論理プログラムを
基礎化した後，clasp を用いて基礎化された論理プロ
グラムの解集合を計算する．基礎化されたルール集合

1 node(1). node(2). node(3). node(4).

2 edge(1,2). edge(2,3). edge(2,4). edge(3,4).

3 col(1). col(2). col(3).

4 1 { color(1,1) ; color(1,2) ; color(1,3) } 1.

5 1 { color(2,1) ; color(2,2) ; color(2,3) } 1.

6 1 { color(3,1) ; color(3,2) ; color(3,3) } 1.

7 1 { color(4,1) ; color(4,2) ; color(4,3) } 1.

8 :- not {color(1,1) ; color(2,1) } 1.

9 :- not {color(1,2) ; color(2,2) } 1.

10 :- not {color(1,3) ; color(2,3) } 1.

11 :- not {color(2,1) ; color(3,1) } 1.

12 :- not {color(2,2) ; color(3,2) } 1.

13 :- not {color(2,3) ; color(3,3) } 1.

14 :- not {color(2,1) ; color(4,1) } 1.

15 :- not {color(2,2) ; color(4,2) } 1.

16 :- not {color(2,3) ; color(4,3) } 1.

17 :- not {color(3,1) ; color(4,1) } 1.

18 :- not {color(3,2) ; color(4,2) } 1.

19 :- not {color(3,3) ; color(4,3) } 1.

コード 4 基礎化された論理プログラム

をコード 4に示す．4–7行目がコード 2の 2行目に，
8–19行目がコード 2の 3行目に対応している．

4 ASPを用いた組合せ遷移問題の解法
ASP 技術を用いた組合せ遷移問題の解法として，
有界組合せ遷移 (Bounded Combinatorial Reconfig-

uration) を提案し，そのソルバーの実装方法につい
て述べる．

4. 1 有界組合せ遷移
有界組合せ遷移では，組合せ遷移問題に対して，制
限された長さの遷移系列を論理プログラムとして記
号的に表現し，その論理プログラムを ASP システ
ムで実行することにより，到達可能性の検査を行う．
ASP システムが充足可能と判定した場合，制限され
た長さの到達可能な遷移系列が存在することを意味
する．逆に充足不能と判定した場合，制限された長さ
では到達不能であることを意味する．この場合，遷
移系列の長さを増加させた論理プログラムを構成し，
再び ASP システムによる実行を繰り返す．有界組合
せ遷移は，到達可能な遷移系列を探すだけで，到達不
能の証明は行わない不完全な手続きであるが，検査す
べき遷移系列の長さに上限が存在する場合には，完全



な手続きとなる．例えば，基の組合せ問題の直径 (任
意の 2つの実行可能解の最短経路のうちの最大長)を
上限とすればよい †3．
組合せ遷移問題が与えられたとき，基の組合せ問
題の変数集合 x = {x1, x2, . . . , xn}，各遷移ステップ
t ≥ 0に対して，ステップ tでの各変数の値を表す変数
集合 xt = {xt

1, x
t
2, . . . , x

t
n} を導入する．ここで，ス

タート状態の制約を表す論理式を S(x)，基の組合せ問
題の制約を表す論理式をC(x)，遷移制約を表す論理式
を T (x,x′)，ゴール状態の制約を表す論理式を G(x)

とする．スタート状態から ℓステップ遷移した後の各変
数の値xℓは，S(x0)∧

∧ℓ
t=0 C(xt)∧

∧ℓ
t=1 T (x

t−1,xt)

を満たす．そこで，ゴール状態の制約を満たすかどう
か判定するため，G(xℓ)を付け加えた論理式 φℓ を構
成する．

φℓ = S(x0)

∧
ℓ∧

t=0

C(xt) ∧
ℓ∧

t=1

T (xt−1,xt)

∧G(xℓ)

与えられたステップ長 ℓに対し φℓ が充足可能であれ
ば，到達可能な遷移系列が得られる．

4. 2 有界組合せ遷移の基本ソルバー
ASP システムを用いて有界組合せ遷移を行う手続
きは以下のようになる．
1. ステップ長 ℓを ℓ = 0とする．
2. φℓ を論理プログラムとして記述し，ファクト形
式の CoRe問題インスタンスとともに，ASP シ
ステムに入力として与える．

3. ASP システムの出力結果が充足可能であれば
終了する．充足不能であれば ℓ の値を 1 増加さ
せ，2に戻り手続きを繰り返す．ただし，ℓが基
の問題の直径を超えたところで繰り返しを停止
できる．

この手続きは，与えられた組合せ遷移問題に対して，
到達可能な最短の遷移系列を探索する．一般に，到達

†3 理論的には，問題の直径を超えるまで手続きを繰り返
せばよいが，実用的な問題ではしばしば直径が大きす
ぎ，現実的には適用できない場合もある．

1 n(4). e(4).

2 node(1). node(2). node(3). node(4).

3 edge(1,2). edge(2,3). edge(2,4). edge(3,4).

4 start(1,4). start(2,2). start(3,3). start(4,1).

5 goal(1,3). goal(2,4). goal(3,2). goal(4,1).

コード 5 k = 4 彩色遷移問題のファクト表現

(graph_reconfig.lp)

不能の証明は行えない不完全な手続きであるが，検査
すべきステップ長 ℓ に上限が存在する場合には完全
な手続きとなり，到達不能も判定できる．この手続き
を，単純に実装したプログラムを基本ソルバーとよぶ
ことにする．
k彩色遷移問題を例にとって，基本ソルバーを用い
た問題解法を説明する．図 2の k = 4彩色遷移問題の
ASP ファクト表現をコード 5に示す．コード 1のグ
ラフ点彩色問題のファクト表現と比較して，スタート
状態を表す start/2 とゴール状態を表す goal/2 が
追加されている．ここで，色 1 は赤，色 2 は青，色
3は黄，色 4は緑とする．例えば，start(1,4).は，
スタート状態 (t = 0)において頂点 1 が色 4 (緑)で
塗られることを表す．
コード 6に，k=c彩色遷移問題を解く論理プログラ
ムとして changed符号化を示す．コード中の c は色
数，length はステップ長を表す．1行目のアトム t/1

はステップ数を表す．アトム color(X,C,T) は，ス
テップ Tにおいて，頂点 Xが色 Cで塗られることを意
味する．3行目はスタート状態の制約を表す．5–7行
目は基のグラフ点彩色問題の制約を表す．9–11行目は
遷移制約を表す．9行目の補助アトム changed(X,T)

は，ステップ T-1と Tの間で頂点 Xの色が変化した
ことを意味する．11行目のルールは，個数制約を用
いて「各ステップ T において色が変化する頂点はた
だ 1つのみ」という制約を表している．
コード 7 に，遷移制約の表現方法が異なる

unchanged 符号化を示す．changed 符号化との違い
は，9–10行目のルールだけである．9行目の補助ア
トム unchanged(X,T)は，ステップ T-1と Tの間で
頂点 Xの色が変化しないことを意味する．10行目の
ルールは，個数制約を用いて「各ステップ Tにおいて



1 col(1..c). t(0..length).

2 % start

3 :- not color(X,C,0), start(X,C).

4 % constraints

5 1 { color(X,C,T): col(C) } 1 :- node(X), t(T).

6 :- not { color(X,C,T); color(Y,C,T) } 1,

7 edge(X,Y), col(C), t(T).

8 % transition

9 changed(X,T) :- color(X,C1,T), color(X,C2,T-1),

10 C1 != C2, T >= 1.

11 :- not 1 { changed(X,T) } 1, t(T), T >= 1.

12 % goal

13 :- not color(X,C,length), goal(X,C).

コード 6 k =c 彩色遷移問題を解く changed符号化

(changed.lp)．定数 length はステップ長を表す．

1 col(1..c). t(0..length).

2 % start

3 :- not color(X,C,0), start(X,C).

4 % constraints

5 1 { color(X,C,T): col(C) } 1 :- node(X), t(T).

6 :- not { color(X,C,T); color(Y,C,T) } 1,

7 edge(X, Y), col(C), t(T).

8 % transition

9 unchanged(X,T) :- color(X,C,T), color(X,C,T-1), T >= 1.

10 :- not N-1 { unchanged(X,T) } N-1 , n(N), t(T), T >= 1.

11 % goal

12 :- not color(X,C,length), goal(X,C).

コード 7 k =c 彩色遷移問題を解く unchanged符号化

(unchanged.lp)．定数 length はステップ長を表す．

色が変化しない頂点は N-1個」という制約を表して
いる．この unchanged符号化の特徴は，changed符
号化と比較して，基礎化後のルール数が少なく抑えら
れる点である．
基本ソルバーは，CoRe問題インスタンスのファク
ト表現と，CoRe問題を解く符号化を入力として，到
達可能性を検査する．コード 8 に，k 彩色遷移問題
に対する基本ソルバーの動作を示す．この例より，ス
テップ長 lengthを 0から 3まで変化させ，clingo を
合計 4 回実行することにより，最短の遷移系列が得
られることがわかる．length=3で得られた解集合は，
図 2に示した遷移系列に対応している．

% clingo graph_reconfig.lp changed.lp -c c=4 -c length=0

clingo version 5.4.0

Reading from graph_reconfig.lp ...

Solving...

UNSATISFIABLE

Models : 0

Calls : 1

CPU Time : 0.003s

% clingo graph_reconfig.lp changed.lp -c c=4 -c length=1

clingo version 5.4.0

Reading from graph_reconfig.lp ...

Solving...

UNSATISFIABLE

Models : 0

Calls : 1

CPU Time : 0.003s

% clingo graph_reconfig.lp changed.lp -c c=4 -c length=2

clingo version 5.4.0

Reading from graph_reconfig.lp ...

Solving...

UNSATISFIABLE

Models : 0

Calls : 1

CPU Time : 0.005s

% clingo graph_reconfig.lp changed.lp -c c=4 -c length=3

clingo version 5.4.0

Reading from graph_reconfig.lp ...

Solving...

Answer: 1

color(1,4,0) color(2,2,0) color(3,3,0) color(4,1,0)

color(1,3,1) color(2,2,1) color(3,3,1) color(4,1,1)

color(1,3,2) color(2,4,2) color(3,3,2) color(4,1,2)

color(1,3,3) color(2,4,3) color(3,2,3) color(4,1,3)...

SATISFIABLE

Models : 1+

Calls : 1

CPU Time : 0.005s

コード 8 clingo の実行例

4. 3 ソルバーの改良
有界組合せ遷移は，ステップ長 ℓを増やしながら，
複数の φℓ を繰り返し解く必要がある．しかし，各 φℓ

に含まれる制約の大部分は同じであるため，実際に

図 3 インクリメンタル ASP 解法を用いた改良ソルバーの構成図

φℓ−1 と φℓ の差をみてみると，
φℓ−1 ⊕ φℓ = {C(xℓ), T (xℓ−1,xℓ), G(xℓ−1), G(xℓ)}
であり，多くの制約は共通していることがわかる．そ
のため，各 φℓ に対して毎回 ASP システムを実行す
る基本ソルバーは，同一の探索空間を何度も調べるこ
とになり求解効率が低下する．また，毎回 φℓ 全体を
基礎化するオーバーヘッドも無視できない．
この問題を解決するために，有界モデル検査やプラ
ンニングの実装で使われるインクリメンタル ASP 解
法を用いる [8]．この解法は，ルールを追加・削除し
た一連の論理プログラムを，同様の探索失敗を防ぐた
めに獲得した学習節を再利用しながら，連続的に解く
ための手法である．
インクリメンタル ASP 解法を用いた改良ソルバー
は，ステップ長 ℓを増やしながら，段階的に φℓ を構
成し，獲得した学習節を再利用した効率のよい到達可
能性の検査を可能とする．改良ソルバーの構成を図 3

に示す．
改良ソルバーにおける有界組合せ遷移を行う手続
き (図 3の core.lp)は以下のようになる．
1. ASP システムを起動する．
2. ファクト形式の CoRe 問題インスタンスと，S

を表すサブプログラム，C と T を表すサブプロ
グラム，G を表すサブプログラムから構成され
る論理プログラムを与える．

3. ステップ長 ℓを ℓ = 0とする．
4. ℓ > 0であれば，G(xℓ−1)を表すルールを削除
する．

5. ℓ = 0 であれば，S(x0) を表すルールを追加
する．

6. C(xℓ)と T (xℓ−1,xℓ)を表すルールを追加する．
7. G(xℓ)を表すルールを追加する．
8. ASP システムを用いて到達可能性を検査する．

9. ASP システムの出力結果が充足可能であれば終
了する．充足不能であれば ℓの値を増加させ，4

に戻り手続きを繰り返す．ただし，ℓが基の問題
の直径を超えたところで繰り返しを停止できる．

10. ASP システムを終了する．
到達不能の証明については，4. 2節で述べた基本ソル
バーの手続きと同じである．
図 4に，clingoの Pythonインターフェースを用い
て上記の手続きを実装した改良ソルバーを示す．変数
core_maxはステップ長 ℓの上限値を表す．core_stop
はプログラムの停止条件を表す変数であり，基本的に
は SAT を指定する．11 行目で定義されている step

は ℓ に対応する．16–18 行目は手続き 4 に対応して
いる．19–20行目は手続き 5に対応している．21行
目は手続き 6に対応している．22行目は手続き 7に
対応している．23行目でステップ数 ℓで追加される
ルールの基礎化を行い，25行目の prg.solve()で到
達可能性を検査している．
改良ソルバーの入力となる論理プログラムは，base,

step(t), check(t) の 3つのパートから構成される．
base パートには，スタート状態の制約 S を記述す
る．step(t) パートには，各ステップ 0 ≤ t ≤ ℓ に
おいて満たすべき基の組合せ問題の制約 C および遷
移制約 T を記述する．check(t) パートには，ゴール
状態の制約 Gを記述する．コード 9とコード 10に，
k=c彩色遷移問題を解く 2種類の論理プログラムを
示す．これらは，それぞれコード 6とコード 7 の論
理プログラムに対応している．改良ソルバーの実行例
をコード 11に示す．得られた解集合は，図 2の遷移
系列に対応している．



1 #script (python)

2 import clingo

3

4 def get(val, default):

5 return val if val != None else default

6

7 def main(prg):

8 core_stop = get(prg.get_const("core_stop"), clingo.String("SAT"))

9 core_max = prg.get_const("core_max")

10

11 step, ret = 0, None

12 while ((core_max is None or step <= core_max.number) and

13 (step == 0 or (core_stop.string == "SAT" and not ret.satisfiable) or

14 (core_stop.string == "UNSAT" and not ret.unsatisfiable))):

15 parts = []

16 if step > 0:

17 prg.release_external(clingo.Function("query", [step-1]))

18 prg.cleanup()

19 else:

20 parts.append(("base", []))

21 parts.append(("step", [step]))

22 parts.append(("check", [step]))

23 prg.ground(parts)

24 prg.assign_external(clingo.Function("query", [step]), True)

25 ret, step = prg.solve(), step+1

26

27 print(’c step: %s’ % str(step-1))

28 if (step != 0 and

29 (core_stop.string != "SAT" or ret.satisfiable) and

30 (core_stop.string != "UNSAT" or ret.unsatisfiable)):

31 print(’s REACHABLE’)

32 else:

33 print(’s UNREACHABLE’)

34

35 #end.

36 #program check(t).

37 #external query(t).

図 4 clingo の Python API を用いた改良ソルバーの実装 (core.lp)

5 実行実験
提案した有界組合せ遷移の有効性を評価するため
に，k彩色遷移問題をベンチマークとして開発したソ
ルバーの性能評価を行った．ベンチマーク問題には，
独自に作成した k 彩色遷移問題 (90問)を使用した．
作成方法は以下の通りである．
• Graph Coloring and its Generalization †4 で公
開されているグラフ点彩色問題のインスタンス

†4 https://mat.tepper.cmu.edu/COLOR04/

127個のうち，彩色数が判明している 44個 [15]

をベンチマーク問題の候補とした．
• 44個のグラフに対して，彩色数での実行可能解
の総数を計算できるか調査した．総数が得られた
9個のグラフを表 2に示す．
• 総数が得られた 9個のグラフについて，実行可
能解をランダムに 2 つ選びベンチマーク問題を
各 10問ずつ作成した (計 90問)．

k彩色遷移問題を解く論理プログラムには，changed
符号化，unchanged 符号化，changed_inc 符号化，
unchanged_inc符号化の 4つを使用した．ステップ



1 #program base.

2 col(1..c).

3 color(X,C,0) :- start(X,C).

4

5 #program step(t).

6 1 { color(X,C,t): col(C) } 1 :- node(X).

7 :- not { color(X,C,t); color(Y,C,t) } 1, edge(X,Y), col(C).

8

9 changed(X,t) :- color(X,C1,t), color(X,C2,t-1),

10 C1 != C2, t >= 1.

11 :- not 1 { changed(X,t) } 1, t >= 1.

12

13 #program check(t).

14 :- not color(X,C,t), goal(X,C), query(t).

コード 9 k=c 彩色遷移問題を解く changed_inc 符号化

(changed_inc.lp)

1 #program base.

2 col(1..c).

3 :- not color(X, C, 0), start(X, C).

4

5 #program step(t).

6 1 { color(X,C,t): col(C) } 1 :- node(X).

7 :- not { color(X,C,t); color(Y,C,t) } 1, edge(X,Y), col(C).

8

9 unchanged(X,t) :- color(X,C,t), color(X,C,t-1), t >= 1.

10 :- not N-1 { unchanged(X,t) } N-1 , n(N), t >= 1.

11

12 #program check(t).

13 :- not color(X,C,t), goal(X,C), query(t).

コード 10 k=c 彩色遷移問題を解く unchanged_inc 符

号化 (unchanged_inc.lp)

長の上限値には，解の総数から 1 ひいた値を使用し
た †5．ASP システムには，clingoのバージョン 5.4.0

(jumpy) を使用した．計算環境は Mac OS 3.2GHz 6

コア Intel Core i7，64GBメモリである．1問あたり
の制限時間は 1時間とした．
表 3に実験結果を示す．左から順に問題名，到達可
能/到達不能の別，ステップ長，各符号化の CPU 時
間を表している．解けた問題の数は，changed符号化

†5 ただし，グラフ myciel4 から作成した問題について
は，clingo が対応している数よりも解の総数が大き
かったため，ステップ長の上限値を 231 − 1 とした．

% clingo graph_reconfig.lp changed_inc.lp core.lp -c c=4

clingo version 5.4.0

Reading from graph_reconfig.lp ...

Solving...

Solving...

Solving...

Solving...

Answer: 1

color(1,4,0) color(2,2,0) color(3,3,0) color(4,1,0)

color(1,3,1) color(2,2,1) color(3,3,1) color(4,1,1)

color(1,3,2) color(2,4,2) color(3,3,2) color(4,1,2)

color(1,3,3) color(2,4,3) color(3,2,3) color(4,1,3)...

SATISFIABLE

c step: 3

s REACHABLE

Models : 1+

Calls : 4

CPU Time : 0.029s

コード 11 提案ソルバーの実行例

表 2 独自に生成した k 彩色遷移問題のベンチマーク問題

グラフ名 頂点数 辺数 彩色数 解の総数
1-FullIns_3 30 100 4 50,693,280

le450_5a 450 5,714 5 3,840

le450_5c 450 9,803 5 120

le450_5d 450 9,757 5 960

myciel3 11 20 4 12,480

myciel4 23 71 5 2,845,658,400

queen5_5 25 160 5 240

queen6_6 36 290 7 100,800

queen7_7 49 476 7 20,160

が 21問，unchanged符号化が 21問，changed_inc

符号化が 67問，unchanged_inc符号化が 71問であ
る．unchanged_incが最も多くの問題を解き，平均
CPU 時間も最も短い．図 5にカクタスプロットを示
す．図 5の縦軸は CPU時間，横軸は解けた問題数を
表す．図より，unchanged_inc 符号化が，他の符号
化と比較して，より多くの問題をより高速に解いてい
ることが確認できる．以上のことから，インクリメン
タル ASP解法を用いた改良ソルバーの優位性が確認
できた．



表 3 実験結果: CPU 時間
問題名 到達可能/到達不能 ステップ長 changed unchanged changed_inc unchanged_inc

基本ソルバー 改良ソルバー
1-FullIns 3 col4 1 REACHABLE 11 1.374 1.071 1.291 0.756
le450 5a col5 1 UNREACHABLE 3,839 TO TO 3.761 2.991
le450 5a col5 2 UNREACHABLE 3,839 TO TO 2.943 11.673
le450 5a col5 3 UNREACHABLE 3,839 TO TO 4.662 22.048
le450 5a col5 4 UNREACHABLE 3,839 TO TO TO 7.776
le450 5a col5 5 UNREACHABLE 3,839 TO TO TO 26.723
le450 5a col5 6 UNREACHABLE 3,839 TO TO 3.618 3.220
le450 5a col5 7 UNREACHABLE 3,839 TO TO 1318.024 8.698
le450 5a col5 8 UNREACHABLE 3,839 TO TO 3.367 18.226
le450 5a col5 9 UNREACHABLE 3,839 TO TO 2.943 2.847
le450 5a col5 10 UNREACHABLE 3,839 TO TO 2.968 3.196
le450 5c col5 1 UNREACHABLE 119 TO TO 2.189 2.165
le450 5c col5 2 UNREACHABLE 119 TO TO 150.018 2.148
le450 5c col5 3 UNREACHABLE 119 TO TO 150.283 2.701
le450 5c col5 4 UNREACHABLE 119 TO TO 152.434 2.102
le450 5c col5 5 UNREACHABLE 119 TO TO 2.209 2.114
le450 5c col5 6 UNREACHABLE 119 TO TO 2.219 2.141
le450 5c col5 7 UNREACHABLE 119 TO TO 151.252 2.150
le450 5c col5 8 UNREACHABLE 119 TO TO 2.878 2.176
le450 5c col5 9 UNREACHABLE 119 TO TO 2.827 2.146
le450 5c col5 10 UNREACHABLE 119 TO TO 2.246 2.166
le450 5d col5 1 UNREACHABLE 959 TO TO 2.680 5.653
le450 5d col5 2 UNREACHABLE 959 TO TO 4.097 2.621
le450 5d col5 3 UNREACHABLE 959 TO TO 3.421 2.600
le450 5d col5 4 UNREACHABLE 959 TO TO 155.850 6.960
le450 5d col5 5 UNREACHABLE 959 TO TO 3.294 2.618
le450 5d col5 6 UNREACHABLE 959 TO TO 199.336 25.273
le450 5d col5 7 UNREACHABLE 959 TO TO 3.308 12.001
le450 5d col5 8 UNREACHABLE 959 TO TO 2.695 15.224
le450 5d col5 9 UNREACHABLE 959 TO TO 3.351 13.121
le450 5d col5 10 UNREACHABLE 959 TO TO 333.183 7.974
myciel3 col4 1 REACHABLE 11 0.179 0.148 0.065 0.065
myciel3 col4 3 REACHABLE 11 0.206 0.164 0.082 0.080
myciel3 col4 4 REACHABLE 13 0.486 0.445 0.303 0.583
myciel3 col4 7 REACHABLE 8 0.090 0.072 0.042 0.036
myciel3 col4 8 REACHABLE 9 0.119 0.094 0.048 0.044
myciel4 col5 2 REACHABLE 16 42.482 48.721 20.308 134.056
myciel4 col5 3 REACHABLE 14 17.098 21.211 5.286 80.848
myciel4 col5 4 REACHABLE 7 0.256 0.193 0.086 0.072
myciel4 col5 6 REACHABLE 10 1.525 1.391 1.052 1.558
myciel4 col5 10 REACHABLE 17 212.682 73.646 29.999 200.068
queen5 5 col5 1 UNREACHABLE 239 439.863 302.994 0.073 0.082
queen5 5 col5 2 UNREACHABLE 239 437.481 301.194 0.074 0.057
queen5 5 col5 3 UNREACHABLE 239 441.161 303.136 0.073 0.589
queen5 5 col5 4 UNREACHABLE 239 443.299 304.547 0.061 0.057
queen5 5 col5 5 UNREACHABLE 239 441.616 302.366 0.074 0.067
queen5 5 col5 6 UNREACHABLE 239 446.811 301.201 0.061 0.149
queen5 5 col5 7 UNREACHABLE 239 441.071 303.449 0.074 0.056
queen5 5 col5 8 UNREACHABLE 239 442.539 303.821 0.073 0.137
queen5 5 col5 9 UNREACHABLE 239 443.353 302.952 0.074 0.066
queen5 5 col5 10 UNREACHABLE 239 446.027 305.350 0.073 0.058
queen6 6 col7 1 UNREACHABLE 100,799 TO TO 5.375 5.488
queen6 6 col7 2 UNREACHABLE 100,799 TO TO 5.337 5.292
queen6 6 col7 3 UNREACHABLE 100,799 TO TO 5.345 4.390
queen6 6 col7 4 UNREACHABLE 100,799 TO TO 5.714 4.358
queen6 6 col7 5 UNREACHABLE 100,799 TO TO 4.274 5.178
queen6 6 col7 6 UNREACHABLE 100,799 TO TO 5.403 5.129
queen6 6 col7 7 UNREACHABLE 100,799 TO TO 5.219 4.418
queen6 6 col7 8 UNREACHABLE 100,799 TO TO 4.300 4.389
queen6 6 col7 9 UNREACHABLE 100,799 TO TO 5.365 5.407
queen6 6 col7 10 UNREACHABLE 100,799 TO TO TO 4.412
queen7 7 col7 1 UNREACHABLE 20,159 TO TO 2804.714 949.947
queen7 7 col7 2 UNREACHABLE 20,159 TO TO 1.342 907.100
queen7 7 col7 3 UNREACHABLE 20,159 TO TO 1.586 1.499
queen7 7 col7 4 UNREACHABLE 20,159 TO TO TO 2.768
queen7 7 col7 5 UNREACHABLE 20,159 TO TO 1.622 6.052
queen7 7 col7 6 UNREACHABLE 20,159 TO TO 1.803 1626.628
queen7 7 col7 7 UNREACHABLE 20,159 TO TO 1.343 1.590
queen7 7 col7 8 UNREACHABLE 20,159 TO TO 1.481 3.652
queen7 7 col7 9 UNREACHABLE 20,159 TO TO 1224.680 3.760
queen7 7 col7 10 UNREACHABLE 20,159 TO TO 1.557 1.349

平均 CPU 時間 223.796 151.341 101.758 59.095

解けた問題数 21 21 67 71

6 おわりに
本稿では，解集合プログラミング (ASP)を用いた
組合せ遷移問題の解法として，有界組合せ遷移を提
案し，そのソルバーの実装方法について述べた．有界

組合せ遷移の有効性を評価するために，代表的な組
合せ遷移問題の 1つである k 彩色遷移問題をベンチ
マークとして，提案ソルバーの性能評価を行った．そ
の結果，インクリメンタル ASP 解法を用いた改良ソ
ルバーがより多くの問題を高速に解くことができる

ことを確認した．
組合せ遷移問題を解く汎用ソルバーの研究開発は
まだ始まったばかりである．今後の課題として，改良
ソルバーの高速化があげられる．高速化の手法として
は，アルゴリズムの刷新や探索ヒューリスティックス
の改良を考えている．集合被覆問題や独立点集合を基
とする組合せ遷移問題，15パズルなどの問題記述例
の蓄積も重要な課題の一つである．また，k彩色遷移
問題の ASP 符号化の改良も考えている．

